Types and Typechecking

© 1998, 1999 M. Lawford

Outline

• Paradoxes

• Hierarchy of Types

• Sets, Sorts & Types

• Typechecking

• Application:

Correctness of Tabular Specifications

• Summary

Paradoxes

paradox - par·a·dox Etymology: From Greek paradoxon, from neuter of paradoxos contrary to expectation,

• a self-contradictory statement that at first seems true

• an argument that apparently derives selfcontradictory conclusions by valid deduction from acceptable premises

Paradoxes result from self-referential statements. E.g.

Liar’s paradox: The Cretan Epimenides said

All Cretans are liars, and all statements made by Cretans are lies.

Russel’s paradox

Bertrand Russel showed that naive set theory was inconsistent with the following paradox:

Let P be the set of all sets that do not contain themselves as an element. P = {Q ∈ sets|Q ∉ Q} e.g. ∅ ∈ P and {1, 2} ∈ P and {1, 2, {1, 2}} ∈ P

Question: Is P ∈ P?

But by def. of P ∈ P ↔ P ∉ P

i.e. P ∈ P ↔ ¬(P ∈ P)

By defining P we have created a contradition!

Conclusion: Naive set theory is inconsistent. We must eliminate such self-referential definitions to make set theory consistent. 3

Type Theory:

Russel created the theory of types, a new set theory that eliminated contradictions by construction.

How? Define a hierarchy of types (all possible sets). Any well defined set can only have elements from lower set levels.

Therefore P ∈ P is always false! A set cannot contain itself since it can only contain elements from levels lower than itself.

Self-reference prohibited by preventing a type α from containing elements of type {α}

Hierarchy of Types:

The universe U is composed of individuals (Induh-viduals) 1. Lowest level - individuals: e.g. integer 2, “Bob” These are things that are not sets.

2. Next level - sets of individuals: which are of type U → {T, F } E.g. set of integers Z, set of students, etc.

3. Higher levels - Let α and β be types from previous levels. Then α → β is a type. Also α → {T, F } is a type.

E.g. The set of class lists: C : (U → {T, F }) → {T, F } A function f : α → β has type or signature α → β A function’s return type is the range type (e.g. β for f above). 5

Sets, Sorts & Types

For our purposes, a type is just a set. Type α → β denotes the set of all (total) functions from α to β.

E.g. In PVS [[real, nzreal] -> real] is the set of all functions from real × non-zero reals to reals.

/:[[real, nzreal] -> real]

is an instance of type [[real, nzreal] -> real] Some of the more algebraic treatments of logic refer to sorts instead of types. A sort is just a non-empty type.

Typechecking

Typed programming languages can check for easily decidable properties:

• use of undefined terms

• adding a boolean to an integer

• security violations (java)

These are properties that can be check mechanically. A language is type safe if programs exhibiting these properties will be rejected during typechecking (often during compilation).

PVS also automatically identifies these problems in specification files when they are typechecked.

Typechecking in PVS

More general typechecking is needed to make sure that formulas are well typed (i.e. never result in undefined terms).

Predicate subtypes with typechecking can be used to check for:

• division by zero

• out of bound array references

• more complicated properties (e.g. invariant properties of a database system)

Many properties are not effectively decidable (i.e. no general algorithm exists to check them). But we may still be able to prove them!

The use of predicate subtypes allows PVS to automatically generate the proof obligations (TCCs - Type Correctness Conditions) to guarantee formulas are well typed. 8

Predicate Subtypes

In our setting types can be thought of as sets. Thus a type α is a subtype of type β if the defining set of α is a subset of the defining set of β.

Predicate subtypes provide a tightly bound characterization by associating a predicate (property) with a subtype. In PVS, N is a predicate subtype of Z.

nat: NONEMPTY_TYPE = {i:int | i >= 0} CONTAINING 0

The predicate is i ≥ 0.

In the definition of type nzreal, real is the type that will be subtyped and x ≠ 0 is the predicate defining the subtype.

For any P : α → {T, F }, a predicate defined on type α, P defines a subtype, denoted (P ): (P ) = {a ∈ α|P a}

PVS Example

In PVS you can define a predicate: even? : Z → {T, F } Then use it to define predicate subtype of even integers:

even?(i:int):bool =
   EXISTS (j:int): i= 2 * j
even: TYPE = (even?)
f(i:int):even = 2 * i

Here f : Z → even

Interpreted and Uninterpreted Types

Interpreted types such as bool, real etc. provide standard mathematical interpretations.

Uninterpreted types:

• Abstract implementation details

• Allow parametrized types (e.g. sets) that are like C++ templates in LEDA

Example:

class:TYPE
mark:TYPE
transcript:TYPE = set[[class,mark]]

Prelude defines operators and properties of all types of sets using parametrized theory:

sets [T: TYPE]: THEORY
 BEGIN
  set: TYPE = [T -> bool]
  . . .
END sets

11

Empty Sets and Types

Extra care must be taken when dealing with possibly empty sets (types). Consider PVS declaration:

T:TYPE
const:T

declares a constant of type T. Results in following unprovable TCC:

% Existence TCC generated . . . for c: T
  % unfinished
c_TCC1: OBLIGATION (EXISTS (x: T): TRUE);

What’s wrong? By definition c ∈ T but if T = ∅ then we have a contradiction.

This can be fixed by making declaration:

T:NONEMPTY_TYPE
c:T

12

Proving quantified versions for empty and nonempty uninterpreted types.

Dependent types

What? parametrized families of types that can be used to

i) more accurately specify range of function

ii) restrict domain of (subsequent) arguments

Why use dependent types?

• the more specific you can be about a function’s return value the easier it is to prove formulas utilizing it are “well typed” (contain no undefined terms for all possible variable values)

• restricting domain of function arguments w.r.t. current value of previous arguments is only way to make some “functions” total.

How? Make types depend on previous arguments 13

Dependent Types in Function Range

Ex. 1st version of abs(x)

abs(m:real): nonneg_real
  = IF m < 0 THEN -m ELSE m ENDIF

A better version

abs(m:real): {n: nonneg_real | n >= m}
  = IF m < 0 THEN -m ELSE m ENDIF

Note: For abs(x), the range type is dependent on the argument m, providing information in the type that is usually provided through separate lemmas.

h(x:real):nonneg_real=sqrt(abs(x)-x)

1st version generates more TCCs for h. 14

Dependent Types in Function Domain

Ex. Consider √ x − y

  % Dependent Types Example
  sqrt: [nonneg_real -> nonneg_real]

f(x,y:real):nonneg_real=sqrt(x-y)
g(x:real,y:{y:real|x>=y}):nonneg_real=sqrt(x-y)

To see the Type Correctness Conditions generated use the PVS “show-tccs” command:

% Subtype TCC generated for x - y
  % unfinished
f_TCC1: OBLIGATION
  (FORALL (x: real, y: real): x - y >= 0);

% Subtype TCC generated for x - y
  % completed
g_TCC1: OBLIGATION
  (FORALL (x: real, y: {y: real | x >= y}):
x - y >= 0);

15

Type Information in PVS

g_TCC1 :

  |-------
{1} (FORALL (x: real, y: {y: real | x >= y}): x - y >= 0)

Rerunning step: (SKOLEM!)
Skolemizing,
this simplifies to:
g_TCC1 :

  |-------
{1} x!1 - y!1 >= 0

Rerunning step: (TYPEPRED "y!1")
Adding type constraints for y!1,
this simplifies to:
g_TCC1 :

{-1} x!1 >= y!1
  |-------
[1] x!1 - y!1 >= 0

Rerunning step: (ASSERT)
Simplifying, rewriting, and recording with decision procedures,
Q.E.D.

(SKOLEM!) followed by (TYPEPRED “t”) implemented by (SKOLEM-TYPEPRED). 16

Undefined Terms in PVS

Note: In PVS everything must be defined before its first use. E.g. If g were redefined as:

g(y:{y:real|x>=y},x:real):nonneg_real=sqrt(x-y)

PVS would produce the typecheck error:

Expecting an expression
No resolution for x

When defining a function f(x1 : t1, x2 : t2, . . . , xn : tn) : tr tj, the type of xj, may only depend on the values of xi’s where 1 ≤ i < j The return type of the function, tr, may depend upon any or all of the argument values.

PVS Command (REPLACE . . . )

Rule I part (b) Substitution of Equals is implemented by the PVS (REPLACE . . . ) command. −1 ϕ1 −2 ϕ2 . . −n tL = tR 1 ψ1 2 ψ2 . .

(REPLACE -n * LR) =⇒

−1 ϕ1[tR|tL] −2 ϕ2[tR|tL] . −n tL = tR 1 ψ1[tR|tL] 2 ψ2[tR|tL] . −1 ϕ1 −2 ϕ2 . . −n tL = tR 1 ψ1 2 ψ2 . .

(REPLACE -n * RL) =⇒

−1 ϕ1[tL|tR] −2 ϕ2[tL|tR] . −n tL = tR 1 ψ1[tL|tR] 2 ψ2[tL|tR] . Variations of (REPLACE . . . ) command let you replace selected instances of equal terms. 18

PVS Commands (EXPAND “t”)

Rule I(a): (∀x)x = x and all its variations are built into PVS

  x,y: VAR real
  f(x,y):real = x+y
  g(x,y):real = x+y
  Ia: THEOREM f(y,1)=g(y,1)

  |-------
{1} (FORALL (y: real): f(y, 1) = g(y, 1))

Rule? (skolem! )

  |-------
{1} f(y!1, 1) = g(y!1, 1)

Rule? (expand "f")
Expanding the definition of f,

  |-------
{1} (1 + y!1 = g(y!1, 1))
Rule? (expand "g")
Expanding the definition of g,
  |-------
{1} TRUE
which is trivially true.
Q.E.D.

Alternatively use (EXPAND* t1 t2 . . . tn) :

Ia :
  |-------
{1} (FORALL (y: real): f(y, 1) = g(y, 1))
Rule? (expand* "f" "g")
Expanding the definition(s) of (f g),
Q.E.D.

PVS Commands (LIFT-IF)

P4 :

  |-------
{1} FORALL (x: real):
IF x >= 0 THEN sqrt(x) ELSE sqrt(-x) ENDIF = sqrt(abs(x))

Rule? (skolem! )

  |-------
{1} IF x!1 >= 0 THEN sqrt(x!1)
       ELSE sqrt(-x!1) ENDIF = sqrt(abs(x!1))

Rule? (lift-if )
Lifting IF-conditions to the top level,
this simplifies to:

  |-------
{1} IF x!1 >= 0 THEN sqrt(x!1) = sqrt(abs(x!1))
      ELSE sqrt(-x!1) = sqrt(abs(x!1))
      ENDIF

Rule? (expand "abs")
P4 :

  |-------
{1} TRUE

which is trivially true.
Q.E.D.

21

Tabular Specifications of Functions

A function f : T1 × . . . × Tm → Tr may have a tabular representation: f(x1, . . . , xm) = c1 c2 . . . cn e1 e2 . . . en Here each ci is a boolean expression (term) and ei is a term of type Tr. When ci is true f returns ei.

The following are sufficient conditions for the table to properly define a (total) function:

Disjoint: i ≠ j → (ci ∧ cj ↔ ⊥) Complete: (c1 ∨ c2 ∨ . . . ∨ cn) ↔ ⊤ Why? Why are they not necessary?

Example: sign(x) = x < 0 x = 0 x > 0 −1 0 1

PVS COND Construct

COND c1 -> e1, c2 -> e2, . . . cn -> en ENDCOND

PVS treats this the same as:

IF c1 THEN e1 ELSIF c2 THEN e2 . . . ELSIF cn−1 THEN en−1 ELSE en

Therefore to prove properties involving COND statements can use (LIFT-IF) with (SPLIT) or (BDDSIMP). (GRIND) can also handle CONDs. (Why?) 23

Typechecking COND Statements

signs: TYPE = { x: int | x >= -1 & x <= 1}

  sign_cond(x): signs =
    COND
      x<0 -> -1,
      x=0 -> 0,
      x>0 -> 1
    ENDCOND

COND causes PVS to generate Disjointness and Completeness TCCs (proof obligations).

% Disjointness TCC generated for
 % COND x < 0 -> -1, x = 0 -> 0, x > 0 -> 1 ENDCOND
  % unfinished
sign_cond_TCC3: OBLIGATION
      (FORALL (x: int):
 NOT (x < 0 AND x = 0)
 AND NOT (x < 0 AND x > 0)
 AND NOT (x = 0 AND x > 0));

25

% Coverage TCC generated for
% COND x < 0 -> -1, x = 0 -> 0, x > 0 -> 1 ENDCOND
  % unfinished
sign_cond_TCC4: OBLIGATION
(FORALL (x: int): x < 0 OR x = 0 OR x > 0);

PVS Table Construct

Equivalent notation that is translated into PVS COND construct

 sign_htable(x): signs = TABLE
                %-------------------%
                |[ x<0 | x=0 | x>0 ]|
                %-------------------%
                | -1 | 0 | 1 ||
                %-------------------%
ENDTABLE

2 dimensional version is nested CONDs

Example: 2A04 Lab 2

lab2 theory (intolerant version) OK

lab2b theory is lab2 with tolerance - 90+ cases of overlap

lab2d theory (somewhat improved) - gives unprovable sequent

func_TCC11.15.1 :

[-1] (a!1 + b!1 < c!1)
[-2] e(a!1, b!1)
[-3] e(b!1, c!1)
[-4] e(a!1, c!1)
  |-------
[1] e(a!1, 0) & e(b!1, 0) & e(c!1, 0)

Rule?

Theorem CE in lab2d verifies existence of counter example.

lab2e final version w/tolerance - works! 28